选修课策略问题

某学校规定，运筹学专业的学生毕业时必须至少学习过两门数学课、三门运筹学课和两门计算机课。这些课程的编号、名称、学分、所属类别和先修课要求如表 1 所示。那么，毕业时学生最少可以学习这些课程中哪些课程。

如果某个学生既希望选修课程的数量少，又希望所获得的学分多，他可以选修哪些课程？

表 1　课程情况

课程编号	课程名称	学分	所属类别	先修课要求
1	微积分	5	数学	
2	线性代数	4	数学	
3	最优化方法	4	数学；运筹学	微积分；线性代数
4	数据结构	3	数学；计算机	计算机编程
5	应用统计	4	数学；运筹学	微积分；线性代数
6	计算机模拟	3	计算机；运筹学	计算机编程
7	计算机编程	2	计算机	
8	预测理论	2	运筹学	应用统计
9	数学实验	3	运筹学；计算机	微积分；线性代数

模型的建立

1 不考虑学分情形：

记 i=1，2，⋯，9 表示 9 门课程的编号。设 $x_i = 1$ 表示第 i 门课程选修，$x_i = 0$ 表示第 i 门课程不选。问题的目标为选修的课程总数最少，即

$$\min Z = \sum_{i=1}^{9} x_i$$

约束条件包括两个方面：

第一方面是课程数量的约束：

每个人最少要学习 2 门数学课，则

$$x_1 + x_2 + x_3 + x_4 + x_5 \geq 2$$

每个人最少要学习 3 门运筹学课 ，则

$$x_3 + x_5 + x_6 + x_8 + x_9 \geq 3$$

每个人最少要学习 2 门计算机课，则有：

$$x_4 + x_6 + x_7 + x_9 \geq 2$$

第二方面是先修课程的关系约束：

如“数据结构 x_4 ”的先修课程是“计算机编程 x_7 ”，

这意味着如果 $x_4 = 1$，必须 $x_7 = 1$，这个条件可以表示为 $x_4 \le x_7$

（注意当 $x_4 = 0$ 时对 x_7 没有限制）。

这样，所有课程的先修课要求可表为如下的约束

“最优化方法 x_3 ”的先修课是“微积分”和“线性代数”，$x_3 \le x_1$, $x_3 \le x_2$

“数据结构 x_4 ”的先修课程是“计算机编程”， $x_4 \le x_7$

“应用统计 x_5 ”的先修课是“微积分”和“线性代数”， $x_5 \le x_1$, $x_5 \le x_2$

“计算机模拟 x_6 ”的先修课程是“计算机编程”，有： $x_6 \le x_7$

“预测理论 x_8 ”的先修课程是“应用统计”，有： $x_8 \le x_5$

“数学实验 x_9 ”是“微积分”和“线性代数”，有： $x_9 \le x_1$, $x_9 \le x_2$

总的 0-1 规划模型为：

$$\min Z = \sum_{i=1}^{9} x_i$$

$$s.t.\begin{cases} x_1 + x_2 + x_3 + x_4 + x_5 \ge 2 \\ x_3 + x_5 + x_6 + x_8 + x_9 \ge 3 \\ x_4 + x_6 + x_7 + x_9 \ge 2 \\ x_3 \le x_1 \ , x_3 \le x_2 \\ x_4 \le x_7 \\ x_5 \le x_1 \ , x_5 \le x_2 \\ x_6 \le x_7 \\ x_8 \le x_5 \\ x_9 \le x_1 \ , x_9 \le x_2 \\ x_1, x_2, \cdots, x_9 = 0 \text{或} 1 \end{cases}$$

解得：

$x_1 = 1, x_2 = 1, x_3 = 1, x_6 = 1, x_7 = 1, x_9 = 1$ 。

即选修课程为：微积分,线性代数.最优化方法,计算机模拟,计算机编程,数学实验。

LINGO 程序为：

```
model:
sets:
item/1..9/:c,x;
endsets
data:
c=5,4,4,3,4,3,2,2,3;
enddata
min=@sum(item(i):x(i));!课程最少;
x(1)+x(2)+x(3)+x(4)+x(5)>=2;
x(3)+x(5)+x(6)+x(8)+x(9)>=3;
x(4)+x(6)+x(7)+x(9)>=2;
x(3)<=x(1);
x(3)<=x(2);
x(4)<=x(7);
x(5)<=x(1);
x(5)<=x(2);
x(6)<=x(7);
x(8)<=x(5);
x(9)<=x(1);
x(9)<=x(2);
@for(item(i):@bin(x(i)));
end
```

2 考虑学分情形：

当要求学分最多时，设各门课程学分为c_i，则增加学分最大的目标函数为：

$$\max Z = \sum_{i=1}^{9} c_i x_i$$

这样总的双目标 0-1 规划模型为：

$$\min Z_1 = \sum_{i=1}^{9} x_i$$

$$\max Z_2 = \sum_{i=1}^{9} c_i x_i$$

$$s.t.\begin{cases} x_1 + x_2 + x_3 + x_4 + x_5 \geq 2 \\ x_3 + x_5 + x_6 + x_8 + x_9 \geq 3 \\ x_4 + x_6 + x_7 + x_9 \geq 2 \\ x_3 \leq x_1 \text{ },x_3 \leq x_2 \\ x_4 \leq x_7 \\ x_5 \leq x_1 \text{ },x_5 \leq x_2 \\ x_6 \leq x_7 \\ x_8 \leq x_5 \\ x_9 \leq x_1 \text{ },x_9 \leq x_2 \\ x_1, x_2, \cdots, x_9 = 0\text{或}1 \end{cases}$$

当把选修课程指定为 6 门时，对学分最大求最优，解得：

$x_1 = 1, x_2 = 1, x_3 = 1, x_5 = 1, x_7 = 1, x_9 = 1$ 。最大学分为 z=22。

即选修课程为：
微积分,线性代数.最优化方法, 应用统计,计算机编程,数学实验。学分达到 22 分。

LINGO 程序为：

```
model:
sets:
item/1..9/:c,x;
endsets
data:
c=5,4,4,3,4,3,2,2,3;
enddata
max=@sum(item(i):c(i)*x(i));
```

```
@sum(item(i):x(i))=6; !课程为6门;
x(1)+x(2)+x(3)+x(4)+x(5)>=2;
x(3)+x(5)+x(6)+x(8)+x(9)>=3;
x(4)+x(6)+x(7)+x(9)>=2;
x(3)<=x(1);
x(3)<=x(2);
x(4)<=x(7);
x(5)<=x(1);
x(5)<=x(2);
x(6)<=x(7);
x(8)<=x(5);
x(9)<=x(1);
x(9)<=x(2);
@for(item(i):@bin(x(i)));
end
```