钢管和易拉罐下料

1. 钢管下料问题

某钢管零售商从钢管厂进货，将钢管按照顾客的要求切割后售出，从钢管厂进货时得到的原料钢管都是 19m。

(1) 现在一客户需要 50 根 4m、20 根 6m 和 15 根 8m 的钢管。应如何下料最节省？

(2) 零售商如果采用的不同切割模式太多，将会导致生产过程的复杂化，从而增加生产和管理成本，所以该零售商规定采用的不同切割模式不能超过 3 种。此外，该客户除需要（1）中的三种钢管外，还需要 10 根 5m 的钢管。应如何下料最节省。

问题（1）分析与模型建立

1 根 19m 的钢管切割为 4m、6m、8m 的钢管的模式，采用向量(k_1,k_2,k_3)表示。

所有模式相当于求解不等式方程：

$$4k_1+6k_2+8k_3\le 19$$

的整数解。但要求剩余材料$r=19-(4k_1+6k_2+8k_3)<4$。

容易得到所有模式见表 1。

表 1　钢管切割模式

模式	4m	6m	8m	余料(m)
1	4	0	0	3
2	3	1	0	1
3	2	0	1	3
4	0	0	2	3
5	0	3	0	1
6	1	1	1	1
7	1	2	0	3

决策变量　用x_i表示按照第 i 种模式(i=1,2,⋯，7)切割的原料钢管的根数。

决策目标　以切割原料钢管的总根数最少为目标，则有

$$\min\ z_1=\sum_{i=1}^{7}x_i$$

以切割后剩余的总余料最小为目标，设第 i 种模式的余料为r_i米。则由表 1 可得

$$\min\ z_2=\sum_{i=1}^{7}r_ix_i$$

设第 i 种切割模式下 4 米长的钢管a_i根，6 米长的钢管b_i根，8 米长的钢管c_i根，10 米长的钢管d_i根。则约束条件有：

4 米长的钢管至少 50 根，有 $\sum_{i=1}^{7} a_i x_i \geq 50$

6 米长的钢管至少 20 根，有 $\sum_{i=1}^{7} b_i x_i \geq 20$

8 米长的钢管至少 15 根，有 $\sum_{i=1}^{7} c_i x_i \geq 15$

因此模型为：

$$\min\ z_1 = \sum_{i=1}^{7} x_i$$

$$\min\ z_2 = \sum_{i=1}^{7} r_i x_i$$

$$s.t.\begin{cases} \sum_{i=1}^{7} a_i x_i \geq 50 \\ \sum_{i=1}^{7} b_i x_i \geq 20 \\ \sum_{i=1}^{7} c_i x_i \geq 15 \\ x_i \text{取整}, i = 1, 2, \cdots, 7 \end{cases}$$

解得：

$$x_1 = 0, x_2 = 15, x_3 = 0, x_4 = 5, x_5 = 0, x_6 = 5, x_7 = 0$$

目标值 z1=25，z2=35。

即 15 根钢管采用切割模式 2：3 根 4m，1 根 6m，余料 1m。

5 根钢管采用切割模式 4：2 根 8m，余料 3m。

5 根钢管采用切割模式 6：1 根 4m，1 根 6m，1 根 8m，余料 1m。

切割模式采用了 3 种，使用钢管 25 根，余料为 35m。

固定 z1=25，求 z2 最小结果一样。说明在该问题中当使用钢管数最少时，余料也最少。

LINGO 程序：

```
model:
sets:
model/1..7/:a,b,c,r,x;
endsets
data:
a=4,3,2,0,0,1,1;
b=0,1,0,0,3,1,2;
c=0,0,1,2,0,1,0;
```

```
r=3,1,3,3,1,1,3;
enddata

min=z1;
z1=@sum(model(i):x(i));!钢管总数;
z2=@sum(model(i):r(i)*x(i));!余料;
@sum(model(i):a(i)*x(i))>=50;!4米长钢管约束;
@sum(model(i):b(i)*x(i))>=20;!6米长钢管约束;
@sum(model(i):c(i)*x(i))>=15;!8米长钢管约束;
@for(model(i):@gin(x(i)));
end
```

问题（2）模型建立

首先分析 1 根 19m 的钢管切割为 4m、6m、8m、5m 的钢管的模式，所有模式相当于求解不等式方程：

$$4k_1+6k_2+8k_3+5k_4 \le 19$$

的整数解。但要求剩余材料 $r=19-(4k_1+6k_2+8k_3+5k_4)<4$ 。

利用 Matlab 程序求出所有模式见表 2。

求出所有模式的 Matlab 程序：

```
number=0;
for k1=0:4
    for k2=0:3
        for k3=0:2
            for k4=0:3
r=19-(4*k1+6*k2+8*k3+5*k4);
if(r>=0)&&(r<4)
    number=number+1;
    fprintf('%2d    %2d    %2d    %2d    %2d %2d\n',number,k1,k2,k3,k4,r);
end
            end
        end
    end
end
```

表 2　钢管切割模式

模式	4m	6m	8m	5m	余料(m)
1	0	0	1	2	1
2	0	0	2	0	3
3	0	1	0	2	3
4	0	1	1	1	0
5	0	2	0	1	2
6	0	3	0	0	1
7	1	0	0	3	0
8	1	0	1	1	2
9	1	1	1	0	1
10	1	2	0	0	3
11	2	0	0	2	1
12	2	0	1	0	3
13	2	1	0	1	0
14	3	0	0	1	2
15	3	1	0	0	1
16	4	0	0	0	3

决策变量　用 x_i 表示按照第 i 种模式(i=1,2,⋯，16)切割的原料钢管的根数。

决策目标　以切割原料钢管的总根数最少为目标，则有

$$\min z_1 = \sum_{i=1}^{16} x_i$$

以切割后剩余的总余料量最小为目标，设第 i 种模式的余料为 r_i 米。则由表 5.6 可得

$$\min z_2 = \sum_{i=1}^{16} r_i x_i$$

设第 i 种切割模式下 4 米长的钢管 a_i 根，6 米长的钢管 b_i 根，8 米长的钢管 c_i 根，5 米长的钢管 d_i 根。则约束条件有：

4 米长的钢管至少 50 根，有　$\sum_{i=1}^{16} a_i x_i \geq 50$

6 米长的钢管至少 20 根，有　$\sum_{i=1}^{16} b_i x_i \geq 20$

8 米长的钢管至少 15 根，有　$\sum_{i=1}^{16} c_i x_i \geq 15$

5 米长的钢管至少 10 根，有 $\sum_{i=1}^{16} d_i x_i \geq 10$

最多使用 3 种切割模式，增设 0-1 变量 $y_i, i = 1, 2, \cdots, 16$ 。

当 $y_i = 0$ 时， $x_i = 0$ ，表示不使用第 i 种切割模式；

当 $y_i = 1$ 时， $x_i \geq 1$ ，表示使用第 i 种切割模式。

因此有： $x_i \geq y_i$ ， $x_i \leq M.y_i$ ， $i = 1, 2, \cdots, 16$

其中 M 足够大，如这里取 100。

$$\sum_{i=1}^{16} y_i \leq 3$$

因此模型为：

$$\min z_1 = \sum_{i=1}^{16} x_i$$

$$\min z_2 = \sum_{i=1}^{16} r_i . x_i$$

$$s.t. \begin{cases} \sum_{i=1}^{16} a_i x_i \geq 50 \qquad \sum_{i=1}^{16} b_i x_i \geq 20 \\ \sum_{i=1}^{16} c_i x_i \geq 15 \qquad \sum_{i=1}^{16} d_i x_i \geq 10 \\ x_i \leq M.y_i, i = 1, 2, \cdots, 16 \\ x_i \geq y_i, i = 1, 2, \cdots, 16 \\ \sum_{i=1}^{16} y_i \leq 3 \\ x_i \text{取整}, i = 1, 2, \cdots, 16 \\ y_i = 0 \text{或} 1, i = 1, 2, \cdots, 16 \\ M \text{足大} \end{cases}$$

解得：

1）当使用钢管数 z1 最小时，求得的解为：

$x_2 = 8, x_{13} = 10, x_{15} = 10$ ，其余为 0。

目标值 z1=28，z2=34。

即 8 根钢管采用切割模式 2：2 根 8m，余料 3m。

10 根钢管采用切割模式 13：2 根 4m，1 根 6m，1 根 5m，余料为 0。

10 根钢管采用切割模式 15：3 根 4m，1 根 6m，余料 1m。

切割模式采用了 3 种，使用钢管 z2=28 根，余料为 z1=34m。

固定 z1=28，求 z2 最小结果一样。说明在该问题中当使用钢管数最少时，余料也最少。

LINGO 程序为：

```
 model:
sets:
model/1..16/:a,b,c,d,r,x,y;
endsets
data:
a=0,0,0,0,0,0,1,1,1,1,2,2,2,3,3,4;
b=0,0,1,1,2,3,0,0,1,2,0,0,1,0,1,0;
c=1,2,0,1,0,0,0,1,1,0,0,1,0,0,0,0;
d=2,0,2,1,1,0,3,1,0,0,2,0,1,1,0,0;
r=1,3,3,0,2,1,0,2,1,3,1,3,0,2,1,3;
enddata

min=z1;
z1=@sum(model(i):x(i));!钢管总数;
z2=@sum(model(i):r(i)*x(i));!余料;
@sum(model(i):a(i)*x(i))>=50;!4米长钢管约束;
@sum(model(i):b(i)*x(i))>=20;!6米长钢管约束;
@sum(model(i):c(i)*x(i))>=15;!8米长钢管约束;
@sum(model(i):d(i)*x(i))>=10;!5米长钢管约束;
@for(model(i):x(i)>=y(i));
@for(model(i):x(i)<=1000*y(i));
@sum(model(i):y(i))<=3;
@for(model(i):@gin(x(i)));
@for(model(i):@bin(y(i)));
end
```

2．易拉罐下料问题

某公司采用一套冲压设备生产一种罐装饮料的易拉罐，这种易拉罐是用镀锡板冲压制成的。易拉罐为圆柱形，包括罐身、上盖和下底，罐身高 10cm，上盖和下底的直径均为 5cm。该公司使用两种不同规格的镀锡板原料：规格 1 的镀锡板为正方形，边长 24cm;规格 2 的镀锡板为长方形，长、宽分别为 32cm 和 28cm。由于生产设备和生产工艺的限制，对于规格 1 的镀锡板原料，只可以按照图 1 中的模式 1、模式 2 或模式 3 进行冲压；对于规格 2 的镀锡板原料只能按照模式 4 进行冲压。使用模式 1、模式 2、模式 3、模式 4 进行每次冲压所需要的时间分别为 1.5s、2s、1s、3s.



模式 1　　　　模式 2　　　　模式 3





模式 4

图 1　模式与易拉罐示意图

该工厂每周工作 40 小时，每周可供使用的规格 1、规格 2 的镀锡板原料分别为 5 万张和 2 万张。目前每只易拉罐的利润为 0.10 元，原料余料损失为 0.001 元/厘米 ^{2}（如果周末有罐身、上盖或下底不能配套组装成易拉罐出售，也看作是原料余料损失）。

问工厂应如何安排每周的生产？

分析与解答：

可计算出每种模式下的余料，罐身面积，罐底或盖面积。

罐身面积　$s_1 = \pi \times 5 \times 10 = 157.08\text{cm}^2$

罐底或盖面积　$s_2 = \pi \times 5^2 / 4 = 19.64\text{cm}^2$

模式 1 的余料：$r_1 = 24 \times 24 - s_1 - 10 \times s_2 = 222.57\text{cm}^2$

模式 2 的余料：$r_2 = 24 \times 24 - 2 \times s_1 - 4 \times s_2 = 183.3\text{cm}^2$

模式 3 的余料：$r_3 = 24 \times 24 - 16 \times s_2 = 261.84\text{cm}^2$

模式 4 的余料：$r_4 = 32 \times 28 - 4 \times s_1 - 5 \times s_2 = 169.5\text{cm}^2$

将 4 种冲压模式的特点列于表 3 中。

表 3　　4 种模式

	罐身个数	罐底、罐盖个数	余料损失（cm^2）	冲压时间(s)	变量
模式 1	1	10	222.57	1.5	x_1
模式 2	2	4	183.3	2	x_2
模式 3	0	16	261.84	1	x_3
模式 4	4	5	169.5	3	x_4

模型建立如下：

决策变量　用 x_i 表示按照第 i 种模式的冲压次数（i=1,2,3,4）, k 表示一周生产的易拉罐个数。为计算不能配套组装的罐身和底、盖造成的原料损失，用 L_1 表示不配套的罐身个数，L_2 表示不配套的底、盖个数。x_i 和 k ，L_1, L_2 是整数。

决策目标　假设每周生产的易拉罐能够全部售出，公司每周的销售利润是 $0.1 \times k$ 。原料余料损失包括两部分：4 种冲压模式下的余料损失和不配套的罐身和底、盖造成的原料损失。

则总的余料损失为：

$$0.001 \times (\sum_{i=1}^{4} r_i . x_i + 157.08 L_1 + 19.64 L_2).$$

目标函数为收益最大，即：

$$\max z = 0.1 \times k - 0.001 \times (\sum_{i=1}^{4} r_i . x_i + 157.08 L_1 + 19.64 L_2).$$

约束条件

时间约束：每周工作时间不超过 40 小时=144000s，则有

$$1.5x_1 + 2x_2 + x_3 + 3x_4 \le 144000$$

原料约束：每周可使用的规格 1 的镀锡板原料为 50000 张，则

$$x_1 + x_2 + x_3 \le 50000$$

规格 2 的镀锡板原料 20000 张

$$x_4 \le 20000$$

配套约束：

由模式可知一周生产的罐身个数与易拉罐个数满足： $k \le x_1 + 2x_2 + 4x_4$

一周生产的罐底、盖个数与易拉罐个数满足： $2k \le 10x_1 + 4x_2 + 16x_3 + 5x_4$

则不配套的罐身个数 L_1 满足：

$$L_1 = x_1 + 2x_2 + 4x_4 - k$$

不配套的底、盖个数 L_2 满足：

$$L_2 = 10x_1 + 4x_2 + 16x_3 + 5x_4 - 2k$$

则总的整数线性规划模型为：

$$\max z = 0.1 \times k - 0.001 \times (\sum_{i=1}^{4} r_i.x_i + 157.08L_1 + 19.64L_2).$$

$$s.t.\begin{cases} 1.5x_1 + 2x_2 + x_3 + 3x_4 \le 144000 \\ x_1 + x_2 + x_3 \le 50000 \\ x_4 \le 20000 \\ k \le x_1 + 2x_2 + 4x_4 \\ 2k \le 10x_1 + 4x_2 + 16x_3 + 5x_4 \\ L_1 = x_1 + 2x_2 + 4x_4 - k \\ L_2 = 10x_1 + 4x_2 + 16x_3 + 5x_4 - 2k \\ x_1, x_2, x_3, x_4 \text{取整} \\ k, L_1, L_2 \text{取整} \end{cases}$$

其中 $r_1 = 222.57, r_2 = 183.3, r_3 = 261.84, r_4 = 169.5$

利用 LINGO 解得：

目标值Z=4298. 188元

$x_1 = 7500, x_2 = 36375, x_3 = 0, x_4 = 20000$ 。

$k = 160250, L_1 = 0, L_2 = 0$

说明冲压第 1 种模式的锡板 7500 张，第 2 种模式的锡板 36375 张，第 3 种模式的锡板 0 张。第 4 种模式的锡板 200000 张。则规格 1 的锡板剩余 $50000-(7500+36375)=6125$ 张，规格 2 的锡板无剩余。

生产易拉罐160250个，罐身、罐底和罐盖都无剩余。

总收益为4298.188元。

LINGO 程序为：

```
model:
sets:
model/1..4/:r,x;
endsets
data:
r=222.57,183.3,261.84,169.5;
enddata
max=0.1*k-0.001*(@sum(model(i):x(i)*r(i))+157.08*L1+19.64*L2);
1.5*x(1)+2*x(2)+x(3)+3*x(4)<=144000; !时间约束;
x(1)+x(2)+x(3)<=50000; !规格1的锡板张数约束;
x(4)<=20000;!规格2的锡板张数约束;
k<=x(1)+2*x(2)+4*x(4); !罐身个数满足条件;
2*k<=10*x(1)+4*x(2)+16*x(3)+5*x(4); !罐底、盖个数满足约束;
L1=x(1)+2*x(2)+4*x(4)-k; !不配套的罐身个数;
L2=10*x(1)+4*x(2)+16*x(3)+5*x(4)-2*k; !不配套的罐底、盖个数;
@for(model(i):@gin(x(i)));
@gin(k);@gin(L1);@gin(L2);
end
```