最优组队问题

某车间要参加单位举办的技术操作比赛，比赛设有 5 个单项和一个全能项目（同时参加 5 个单项）

问题 1：如果比赛规定：

（1）每个车间可派 14 人参加比赛，每人至少参赛一项；

（2） 参加比赛的队员中必须有 3 人参加全能比赛，其余队员参加单项比赛，且参加每个单项比赛的队员数不得超过 6 人（不包括全能队员）；

（3）参加全能的队员不能参加单项；

（4）参加单项比赛的队员至多可以参加 3 个单项；

（5） 参加单项比赛的队员得分是其参加项目得分之和，参加全能比赛的队员得分是其参加项目得分和的 4/5，车间的得分是车间所有参赛队员得分之和。

问如何安排参加比赛最好？

表 1：某车间参加岗位技术比赛队员的期望得分

队员	1	2	3	4	5	6	7	8	9	10	11	12	13	14
单项 1	10	1	4	10	5	5	4	6	2	4	8	6	10	9
单项 2	9	5	6	4	4	7	4	7	8	6	7	8	1	4
单项 3	7	5	5	6	7	7	8	8	7	10	2	6	4	5
单项 4	3	5	9	5	8	6	9	10	6	6	5	4	2	4
单项 5	3	10	8	2	8	7	7	5	8	6	9	8	3	7

模型建立：

建立决策变量:

考察参加全能比赛的情况，令 $y_j = 1$ 表示第 j 人参加全能比赛；

$y_j = 0$ 表示第 j 人不参加全能比赛。

参加全能比赛的为 3 人，则有：

$$\sum_{j=1}^{14} y_j = 3$$

考察参加单项比赛的情况，令 $x_{ij}=1$ 表示第 j 人参加第 i 项单项比赛；

$x_{ij}=0$ 表示第 j 人不参加第 i 项单项比赛。

每个参加全能比赛的不参加单项比赛，则有：

$$x_{ij} \le 1-y_j \qquad j=1,2,\cdots,14; i=1,\cdots,5$$

该约束表示当第 j 人参加全能比赛时（$y_j=1$）， 则第 j 人则不参加所有单项比赛，因此必有 $x_{ij}=0$，$i=1,\cdots,5$。当第 j 人不参加参加全能比赛时（$y_j=0$）， 则第 j 人可参加也可不参加单项比赛，$x_{ij}=0$ 或 1。这样就把参加单项比赛和参加全能比赛分开考虑，计算各自得分。最后考虑总得分，方便建立模型和计算。

参加单项比赛的人至少参加 1 项，我们要求

当 $y_j=0$ 时，要求 $\sum_{i=1}^{5} x_{ij} \ge 1$，则有： $\sum_{i=1}^{5} x_{ij} \ge 1-y_j \qquad j=1,\cdots,14$

上式中当 $y_j=1$ 时，有 $\sum_{i=1}^{5} x_{ij} \ge 0$。

参加单项比赛的人至多参加 3 项，则有：$\sum_{i=1}^{5} x_{ij} \le 3 \qquad j=1,\cdots,14$

对参加单项比赛的队员，每个项目最多允许 6 人参加，则有：

$$\sum_{j=1}^{14} x_{ij} \le 6 \qquad i=1,\cdots,5$$

设 a_{ij} 表示第 j 人参加第 i 个项目的期望得分，该数据已知。

设 s_j 表示第 j 人参加 5 个项目的总得分。则有：

$$s_j=\sum_{i=1}^{5} a_{ij} \quad j=1,2,\cdots,14$$

参加全能比赛的队员总得分为： $z_1=0.8\sum_{j=1}^{14} s_j.y_j$

参加单项比赛的队员总得分为： $z_2=\sum_{i=1}^{5}\sum_{j=1}^{14} a_{ij}.x_{ij}$

则车间 14 人参加比赛总得分为：

$$z=z_1+z_2=0.8\sum_{j=1}^{14} s_j.y_j+\sum_{i=1}^{5}\sum_{j=1}^{14} a_{ij}.x_{ij}$$

因此我们的目标函数是参加比赛的队员总得分最高，因此目标函数为：

$$z = 0.8\sum_{j=1}^{14} s_j.y_j + \sum_{i=1}^{5}\sum_{j=1}^{14} a_{ij}.x_{ij}$$

总的 0-1 线性规划模型为：

$$z = 0.8\sum_{j=1}^{14} s_j.y_j + \sum_{i=1}^{5}\sum_{j=1}^{14} a_{ij}.x_{ij}$$

$$s.t.\begin{cases} \sum_{j=1}^{14} y_j = 3 \\ x_{ij} \le 1 - y_j \qquad j = 1,2,\cdots,14; i = 1,\cdots,5 \\ \sum_{i=1}^{5} x_{ij} \ge 1 - y_j \qquad j = 1,\cdots,14 \\ \sum_{i=1}^{5} x_{ij} \le 3 \qquad j = 1,\cdots,14 \\ \sum_{j=1}^{14} x_{ij} \le 6 \qquad i = 1,\cdots,5 \\ s_j = \sum_{i=1}^{5} a_{ij} \qquad j = 1,2,\cdots,14 \\ x_{ij} = 0\text{或}1 \qquad j = 1,2,\cdots,14; i = 1,\cdots,5 \\ y_j = 0\text{或}1 \qquad j = 1,2,\cdots,14 \end{cases}$$

Lingo 程序为：

```
model:
sets:
person/1..14/:y,s;
item/1..5/;
assign(item,person):x,A;
endsets
data:
A=10 1 4 10 5 5 4 6 2 4 8 6 10 9
9 5  6  4 4 7 4  7  8 6  7  8 1 4
7 5  5  6 7 7 8  8  7 10 2 6 4 5
3 5  9  5 8 6 9 10 6  6  5 4 2 4
3 10 8 2 8 7 7  5  8  6 9 8 3 7;
enddata
max=z;
z=z1+z2; !车间总得分;
z1=0.8*@sum(person(j):y(j)*s(j)); !参加全能比赛的总得分;
z2=@sum(assign(i,j):a(i,j)*x(i,j)); !参加单项比赛的总得分;
```

```
@for(person(j):@sum(item(i):x(i,j))>=1-y(j));!每个参加单项比赛的队员至少参加1项;
@for(person(j):@sum(item(i):x(i,j))<=3); !每个参加单项比赛的队员最多不超过3项;
@for(item(i):@sum(person(j):x(i,j))<=6); !每个项目最多允许6人参加;
@for(assign(i,j):x(i,j)<=1-y(j)); !参加全能比赛的不能参加单项比赛;
@sum(person(j):y(j))=3; !总共只有3人参加全能比赛;
@for(person(j):s(j)=@sum(item(i):a(i,j))); !每个队员所得分;
@for(assign(i,j):@bin(x(i,j)));
@for(person(j):@bin(y(j)));
end
```

求解结果为：

总得分Z=309.8，其中全能得分Z1=76.8，单项得分Z2=233

$$y_5=1, y_6=1, y_{12}=1$$

表示第5、第6、第12名队员参加全能比赛，

其它11人参加单项比赛情况见表2。

表2　参加单项比赛情况

队员	1	2	3	4	7	8	9	10	11	13	14
单项 1	1	0	0	1	0	1	0	0	1	1	1
单项 2	1	1	1	0	0	0	1	1	1	0	0
单项 3	1	0	0	1	1	1	1	1	0	0	0
单项 4	0	1	1	1	1	1	0	1	0	0	0
单项 5	0	1	1	0	1	0	1	0	1	0	1